10

EREMU BEKTORIALAK

10.1. Eremu bektorialak

▼ VectorPlot[ ] funtzioa

★ **VectorPlot**$[\{v_x, v_y\}, \{x, x_{min}, x_{max}\}, \{y, y_{min}, y_{max}\}]$

Planoko puntu bakoitzean {y , x} eremu bektoriala irudikatzen du

```
Clear["Global`*"]
eremubek = VectorPlot[{y, x}, {x, -3, 3}, {y, -3, 3},
  VectorStyle → RGBColor[0.5, 0, 0.7], VectorScale → Small]
```



```
VectorPlot[{{-x, y}, {y, x}}, {x, -3, 3}, {y, -3, 3}]
```



▼ VectorPlot[ ] funtzioaren aukerak

★ VectorColor eta VectorScale

```
eremubektoriala = VectorPlot[{y, x}, {x, -1, 1},
  {y, -1, 1}, VectorScale → Medium, VectorColorFunction → "Rainbow"]
```



★ **Stream**

```
VectorPlot[{y, x}, {x, -1, 1}, {y, -1, 1}, StreamPoints → 15, StreamStyle → Purple,
 StreamScale → Full, VectorPoints → 12, VectorScale → Medium, VectorStyle → Automatic]
```



★ **Eremuak**

```
VectorPlot[{y, x}, {x, -1, 1}, {y, -1, 1}, RegionFunction → Function[{x, y}, x y < 0]]
```



★ **Eremu bektorialak 3D-n**

```
VectorPlot3D[{x, y, z}, {x, -1, 1}, {y, -1, 1}, {z, -1, 1}]
```



10.2. Ekuazio Diferentzial Arrunten soluzioak

▼ DSolve[ekuazioa, funtzioa, aldagaia]

Ekuazio Diferentzial Arrunteko (EDA) y(x) soluzioa aurkitzen du

★ **Lehen ordenako EDA baten ebazpena**

```
ed1 = y'[x] + 4 * y[x] == 0
```

$4\,y[x] + y'[x] == 0$

```
S1 = DSolve[ed1, y[x], x]
```

$\{\{y[x] \to e^{-4x}\,C[1]\}\}$

★ **Bigarren ordenako EDA baten ebazpena**

```
ed2 = y''[x] - 3 y'[x] + 2 * y[x] == 0
```

$2\,y[x] - 3\,y'[x] + y''[x] == 0$

```
S2 = DSolve[ed2, y[x], x]
```

$\{\{y[x] \to e^{x}\,C[1] + e^{2x}\,C[2]\}\}$

★ **NDsolve[ekuazioa, funtzioa, {x, x_{min}, x_{max}}]**

EDA baten zenbakizko soluzioa aurkitzen du definitutako tartean

```
s = NDSolve[{y'[x] == y[x] Cos[x + y[x]], y[0] == 1}, y, {x, 0, 30}]
```

```
{{y → InterpolatingFunction[{{0., 30.}}, <>]}}
```

```
Plot[Evaluate[y[x] /. s], {x, 0, 30}, PlotRange → All]
```



10.3. Lehen ordenako Ekuazio Diferentzial Arruntak

▼ Emaitza orokorra eta emaitza partikularra

★ EDA baten emaitza orokorra

Lortutako emaitza bat oinarritzat hartuz, lehen ordenako EDA-ren soluzioa izango den funtzioa definituko dugu. Berau parametro baten menpeko soluzio familia bat izango da

```
so[x_, c_] = S1[[1, 1, 2]] /. C[1] → c
```

$c\, e^{-4 x}$

★ Soluzio familia

c parametroari balioak emanez soluzio familia bat sortuko dugu

```
listsol = Table[so[x, c], {c, -4, 4, 1}]
```

$\{-4 e^{-4 x}, -3 e^{-4 x}, -2 e^{-4 x}, -e^{-4 x}, 0, e^{-4 x}, 2 e^{-4 x}, 3 e^{-4 x}, 4 e^{-4 x}\}$

```
famsol = Plot[Evaluate[listsol], {x, 0, 1},
  PlotStyle → Thickness[0.005], PlotRange → {-4, 4}]
```



★ Hasierako baldintzadun EDA baten soluzioa

Adibidez, y(0)=2 hasierako baldintzadun ondoko EDA-ren soluzioa aurkitu nahi bada, pauso hauek jarraitzen dira:

```
ed1 = y'[x] + 4 * y[x] == 0
```

$4 y[x] + y'[x] == 0$

```
solo1 = DSolve[{ed1, y[x0] == y0}, y[x], x]
```

$\{\{y[x] \to e^{-4 x+4 x0}\, y0\}\}$

```
yg[x_] = solo1[[1, 1, 2]]
```

$e^{-4 x+4 x0}\, y0$

```
yg[x] /. {x0 → 0, y0 → 2}
```

$2\,e^{-4x}$

```
sp = DSolve[{ed1, y[0] == 2}, y[x], x]
```

$\{\{y[x] \to 2\,e^{-4x}\}\}$

```
yp[x_] = sp[[1, 1, 2]]
```

$2\,e^{-4x}$

Adibidez, y(0)=2 eta y'(0)=1 hasierako baldintzadun ondoko EDA-ren soluzioa aurkitu nahi bada, pauso hauek jarraitzen dira:

```
ed2 = y''[x] - 3 y'[x] + 2 * y[x] == 0
```

$2\,y[x] - 3\,y'[x] + y''[x] == 0$

```
sp2 = DSolve[{ed2, y[0] == 2, y'[0] == 1}, y[x], x]
```

$\{\{y[x] \to -e^{x}\,(-3 + e^{x})\}\}$

```
yp2[x_] = sp2[[1, 1, 2]]
```

$-e^{x}\,(-3 + e^{x})$

▼ Berezitasunak

Adibidez, x^2*y'[x]+y[x]^2==0 EDA-rentzat, soluzio orokor bat oinarritzat hartuz soluzio familia bat sortzen dugu

```
S = DSolve[x^2 * y'[x] + y[x]^2 == 0, y[x], x]
```

$$\left\{\left\{y[x] \to -\frac{x}{1 + x\,C[1]}\right\}\right\}$$

```
solor1[c_, x_] = S[[1, 1, 2]] /. C[1] → c
```

$$-\frac{x}{1 + c\,x}$$

```
listsol = Table[solor1[c, x], {c, -2, 2, .5}]
```

$$\left\{-\frac{x}{1 - 2.\,x}, -\frac{x}{1 - 1.5\,x}, -\frac{x}{1 - 1.\,x}, -\frac{x}{1 - 0.5\,x}, -1.\,x, -\frac{x}{1 + 0.5\,x}, -\frac{x}{1 + 1.\,x}, -\frac{x}{1 + 1.5\,x}, -\frac{x}{1 + 2.\,x}\right\}$$

```
familiasol = Plot[Evaluate[listsol], {x, -2, 3}]
```



Hasierako baldintzadun {x^2*y'[x]+y[x]^2==0,y[0]==2} problemaren emaitza DSolve erabilita planteatzen dugunean ez dugu emaitzarik lortzen

```
Sp = DSolve[{x^2 * y'[x] + y[x]^2 == 0, y[0] == 2}, y[x], x]
```

DSolve::bvnul : For some branches of the general solution, the given boundary conditions lead to an empty solution. ≫

{}

Lehendabizi EDA-ren soluzio orokorra bilatzen saiatuz eta ondoren, hasierako baldintzak betearaztea posible egingo duen c-ren balioa, ez dugu ezta emaitzarik lortzen. Gainera, grafikoan ikus daiteke (0,2) puntutik ez dela EDA horren soluziorik igarotzen.

```
S = DSolve[x^2 * y'[x] + y[x]^2 == 0, y[x], x]
```

$$\left\{\left\{y[x] \to -\frac{x}{1 + x\, C[1]}\right\}\right\}$$

```
solor[c_, x_] = S[[1, 1, 2]] /. C[1] → c
```

$$-\frac{x}{1 + c\, x}$$

```
kons = Solve[solor[c, 0] == 2, c]
```

{}

Orain hasierako baldintzadun {x^2*y'[x]+y[x]^2==0,y[0]==0} problemaren emaitza DSolve erabilita planteatzen dugunean infinitu soluzio aurkitzen ditugu. Grafikoan ikus daiteke (0,0) puntutik igarotzen diren hainbat soluzio daudela.

```
Sp = DSolve[{x^2 * y'[x] + y[x]^2 == 0, y[0] == 0}, y[x], x]
```

DSolve::bvnr : For some branches of the general solution, the
given boundary conditions do not restrict the existing freedom in the general solution. ≫

DSolve::bvsing :
Unable to resolve some of the arbitrary constants in the general solution using the given boundary conditions.
It is possible that some of the conditions have been specified at a singular point for the equation. ≫

$$\left\{\left\{y[x] \to -\frac{x}{1 + x\, C[1]}\right\}\right\}$$

10.4. Lehen ordenako EDA bati lotutako eremu bektoriala

▼ Ukitzaileen eremua

y'=f(x,y) EDA-ri lotutako bektore ukitzaileen eremua {1,f(x,y)} adierazpenaren bidez emana dator

★ y'=x ekuazio diferentziala emanik, soluzio orokorra aurkituko dugu eta soluzio familia irudikatuko dugu

```
ed = y'[x] == x
```

y′[x] == x

```
S = DSolve[ed, y[x], x]
```

$$\left\{\left\{y[x] \to \frac{x^2}{2} + C[1]\right\}\right\}$$

```
so[x_, c_] = S[[1, 1, 2]] /. C[1] → c
```

$$c + \frac{x^2}{2}$$

```
listsol = Table[so[x, c], {c, -5, 8, 1}]
```

$$\left\{-5+\frac{x^2}{2},\ -4+\frac{x^2}{2},\ -3+\frac{x^2}{2},\ -2+\frac{x^2}{2},\ -1+\frac{x^2}{2},\ \frac{x^2}{2},\ 1+\frac{x^2}{2},\ 2+\frac{x^2}{2},\ 3+\frac{x^2}{2},\ 4+\frac{x^2}{2},\ 5+\frac{x^2}{2},\ 6+\frac{x^2}{2},\ 7+\frac{x^2}{2},\ 8+\frac{x^2}{2}\right\}$$

```
famsol = Plot[Evaluate[listsol], {x, -5, 5},
  PlotStyle → {Green, Thickness[0.005]}, PlotRange → {-5, 8}]
```



★ **VectorPlot[{v_x, v_y}, {x, x_{min}, x_{max}}, {y, y_{min}, y_{max}}] funtzioa**

y'=x ekuazio diferentziala emanik, honi lotutako bektore ukitzaileen eremua {1,x} adierazpenaz emana dator

```
eremubek = VectorPlot[{1, x}, {x, -3, 3}, {y, -3, 8},
  VectorStyle → RGBColor[0.2, 0.8, 0.3], VectorScale → Small]
```



★ **Aurreko grafiko guztiak elkartuz**

```
Show[{eremubek, famsol}, PlotRange → {{-3, 3}, {-3, 8}}]
```



★ **“stream” aukera**

```
eremubek = VectorPlot[{1, x}, {x, -3, 3}, {y, -3, 8},
  VectorScale → Small, StreamScale → Full, StreamPoints → 15, StreamScale → Full]
```



▼ Ibilbide ortogonalak

y’ = x ekuazio diferentzialaren soluzio kurben familiarekiko ibilbide ortogonalen ekuazio diferentziala y' = -1/x da eta bektore ukitzailea m'= {- f (x, y).1} = {- x,1} da

★ **Ibilbide ortogonalen ekuazio diferentziala ebazten da, soluzio orokorra definitzen da eta kurba familia irudikatzen da**

```
edto = y'[x] == -1 / x
```

$$y'[x] == -\frac{1}{x}$$

```
Sto = DSolve[edto, y[x], x]
```

```
{{y[x] → C[1] - Log[x]}}
```

```
sgto[x_, c_] = Sto[[1, 1, 2]] /. C[1] → c
```

```
c - Log[x]
```

```
listsol1 = Table[sgto[x, c], {c, -5, 8, 1}]
```

```
{-5 - Log[x], -4 - Log[x], -3 - Log[x], -2 - Log[x], -1 - Log[x], -Log[x], 1 - Log[x],
 2 - Log[x], 3 - Log[x], 4 - Log[x], 5 - Log[x], 6 - Log[x], 7 - Log[x], 8 - Log[x]}
```

```
listsol2 = Table[sgto[-x, c], {c, -5, 8, 1}]
```

```
{-5 - Log[-x], -4 - Log[-x], -3 - Log[-x], -2 - Log[-x], -1 - Log[-x], -Log[-x], 1 - Log[-x],
 2 - Log[-x], 3 - Log[-x], 4 - Log[-x], 5 - Log[-x], 6 - Log[-x], 7 - Log[-x], 8 - Log[-x]}
```

```
famsol1 = Plot[Evaluate[listsol1], {x, 0.01, 3},
   PlotStyle → {Blue, Thickness[0.005]}, PlotRange → {-4, 8}];
```

```
famsol2 = Plot[Evaluate[listsol2], {x, -3, -0.01},
   PlotStyle → {Blue, Thickness[0.005]}, PlotRange → {-4, 8}];
```

```
famsolto = Show[{famsol1, famsol2}, PlotRange → {{-3, 3}, {-3, 8}}]
```



y’=x ekuazio diferentzialari lotutako eremu bektorialaren bektore ukitzailea {1,f(x,y)}={1,x} bada, orduan, ibilbide ortogonalena {-x,1} izango da

```
g1 = VectorPlot[{{-x, 1}, {1, x}}, {x, -3, 3}, {y, -3, 8},
  VectorScale → Small, VectorPoints → 12, VectorStyle → {Purple, Blue}]
```



```
g2 = Show[{famsol, famsol1, famsol2}, PlotRange → {{-3, 3}, {-3, 8}}]
```



★ **Kurba familia eta ukitzaileen eremua batera irudikatuko ditugu**

```
Show[{g1, g2}, PlotRange → {{-3, 3}, {-3, 8}}]
```



```
VectorPlot[{{-x, 1}, {1, x}}, {x, -3, 3}, {y, -3, 8}, StreamPoints → 15,
 StreamScale → Full, VectorPoints → 12, VectorScale → Small, VectorStyle → {Purple, Blue}]
```